WANL-PR(LL)-010
June 1967

Westinghouse Astronuclear Laboratory

SYNTHESIS OF CALCULATIONAL METHODS FOR THE DESIGN AND ANALYSIS OF RADIATION SHIELDS FOR NUCLEAR ROCKET SYSTEMS

Contract No. NAS-8-20414
Contract No. DCN-16-28-0029 (IF)

FINAL PROGRESS REPORT

Volume 8

INFORMATION CATEGORY

Unclassified

M Quigley 7/21/67

AUTHORIZED CLASSIFIER DATE

Astronuclear Laboratory

WANL-PR-(LL)-010

VOLUME 8

SYNTHESIS OF CALCULATIONAL METHODS FOR THE DESIGN AND ANALYSIS OF RADIATION SHIELDS FOR NUCLEAR ROCKET SYSTEMS

DAFT

ODD-K ANGULAR FLUX TAPES PROGRAM

by

R. K. Disney

and

S. L. Zeigler

Contract No. NAS-8-20414
CONTRACT NO. DCN-16-28-0029(IF)

ACKNOWLEDGMENTS

The authors wish to acknowledge the efforts of T. M. Jordan and R. G. Soltesz of Westinghouse Astronuclear Laboratory for their guidance in the formation of the DAFT program. Likewise, the authors appreciate the guidance provided by Mr. Henry E. Stern, Deputy Manager, Nuclear and Plasma Physics Division, George C. Marshall Space Flight Center, the technical monitor of the contract.

ABSTRACT

This report is Volume 8 of nine volumes of the final report on "Synthesis of Calculational Methods for the Design and Analysis of Radiation Shields for Nuclear Rocket Systems." Presented in this volume is a description of the DAFT (ODD-K Angular Flux Tape) program.

DAFT, a FORTRAN IV program for the IBM 7094 computer, is part of the "final" design method as described in Volume I. This program is the data processing routine which prepares angular, spatial, and energy distribution data for input to the FASTER Monte Carlo program (Volume 9) from the surface angular leakage flux data of the two-dimensional transport program ODD-K (Volume 6).

The processing of the surface flux data in a form readily usable in the Monte Carlo program, FASTER, is achieved by reducing the surface multigroup angular leakage flux data into a histogram representation of the spatial, angular, and energy flux at the surface of a two dimensional (R,Z) cylindrical reactor mockup. The DAFT program reduces the ODD-K surface angular flux data such that a limited number of areas, each having a defined angular energy flux, at the periphery of a reactor can be input to the FASTER program. Subsequently, the FASTER program can be used to predict the external radiation environment and/or propellant tank heating with a minimum amount of computer time and a maximum of accuracy for a given cost. If this ODD-K, DAFT combination is not used, the FASTER program must be used with volume distributed sources.

TABLE OF CONTENTS

LIST OF ILLUSTRATIONS

LIST OF TABLES

SECTION

1.0 INTRODUCTION

This report is Volume 8 of nine volumes of the final report on "Synthesis of Calculational Methods for the Design and Analysis of Radiation Shields for Nuclear Rocket Systems." Presented in this volume is a description of the DAFT (ODD-K Angular Flux Tape) program.

The DAFT program, which is written in FORTRAN IV language, is a data processing link in the "final" design method provided for the Marshall Space Flight Center (MSFC). A simplified schematic diagram of the "final" design method is shown in Figure 1. This method is fully described in Volume 1 of this report. As shown in the Figure 1, the DAFT program prepares angular flux data for use in the FASTER program. The starting point for the final design method is the POINT program (Volume 2) which prepares cross section and other basic data for use in the transport programs. In the "final" design method (Figure 1), the ODD-K two-dimensional transport program (Volume 6) provides neutron and photon energy fluxes throughout the reactor geometry. The NAGS data processing program (Volume 7) processes those fluxes and calculates neutron and photon radiation levels, and neutron and photon energy sources within the reactor system. These sources can be employed in either the KAP-V program (Volume 4) or the FASTER Monte Carlo program (Volume 9) for obtaining radiation levels at locations external to the reactor system. In addition, the FASTER program can compute heating rate distributions in the liquid hydrogen propellant (in either an on-axis or an off-axis tank) and the radiation level at the payload. Alternately, the DAFT program (Volume 8) can prepare neutron and photon energy and angular dependent fluxes at the reactor surface from the ODD-K program results for use in the FASTER Monte Carlo program.

The DAFT program prepares the data in a form usable in the FASTER (Volume 9) program. The DAFT program reduces the multigroup and angular fluxes at the surface (i.e., bottom, top, and lateral surfaces of an R, Z reactor geometry) into separate spatial, angular, and energy data for a limited number of surface areas. This program can coalesce the surface flux data from an ODD-K problem into as few as three surface area sources, or into as many as the ODD-K problem had for the outer boundary mesh intervals, representing

the bottom, top, and lateral surfaces of the reactor geometry.

The DAFT program uses the variable dimensioning capability of FORTRAN IV and allows general treatment of the discrete ordinate quadrature order (e. g. , S_2, S_4, S_6, S_8, . . .) up to the limit of 14,000 memory core storage locations available for data. Experience indicates that a DAFT problem using 16 groups and 1188 mesh cells (incremental volumes) of data from an S_6 ODD-K problem (36 radial, 33 axial mesh intervals) and reducing them to 11 surface area sources, required only 5,000 of the total available 14,000 storage locations.

Computer running time for the DAFT program is relatively short. The problem described in the previous paragraph required less than two minutes on the IBM 7094 Model II computer.

Section 2 describes the quantities computed by the DAFT program from the two-dimensional transport results. The program logic is briefly discussed in Section 3. Section 4 presents the input data requirements. The DAFT code output format is discussed in Section 5. The FORTRAN IV source program is listed in the Appendix.



Figure 1. Schematic Diagram of the "Final" Design Method

SECTION

2.0 PROGRAM DESCRIPTION

The DAFT program calculates the spatial, angular, and energy distribution of the leakage neutron or photon energy flux at the surface of a two-dimensional reactor geometry. This program utilizes, as input, the angular flux data on a binary data tape generated by the ODD-K program (Volume 6). The multigroup angular flux (neutron or photon) data from ODD-K is reduced by DAFT into separable spatial, angular, and energy distributions which are provided in histogram form for use as input to the FASTER program (Volume 9). The following sections will describe the required angular flux data from the ODD-K program and the calculations performed on this data.

2.1 ANGULAR FLUX DATA

The DAFT program requires as input the angular flux data on a binary tape as calculated by the ODD-K program. The DAFT program is restricted to the following:

1) The angular flux must be from an R, Z ODD-K geometry model.
2) The discrete ordinate quadrature set which defines the discrete directions of the angular flux which must be a rotationally symmetric set.

These restrictions are imposed in the DAFT program since: (1) all data are computed only for an R, Z geometry, and (2) the angular flux is reduced for all surfaces (bottom, top, and lateral surfaces) in a similar fashion.

Surface angular fluxes are obtained from the ODD-K program at the surface midpoint of each mesh cell on the surface of the reactor geometry. These angular fluxes are discrete direction fluxes as obtained during the ODD-K solution for the scalar fluxes. The directions and the weights (solid angle elements on a unit sphere) are determined by the quadrature scheme employed (Volume 6).

In solving for angular fluxes in an ODD-K, R-Z two-dimensional mesh cell description, each mesh cell is a finite volume element of: $\Delta R = R_{i+1} - R_i$, $\Delta Z = Z_{i+1} - Z_i$, and $\Delta\theta = \theta_{k+1} - \theta_k$ as shown in Figure 2. Because of symmetry in the angle, θ, solutions are only calculated in ODD-K at points A, B, C, D, and P. Points A, B, C, and D are midpoints of each surface of the mesh cell. The angular fluxes which are obtained at each of these points are then used to calculate the angular and the scalar flux at the midpoint, P, of the mesh cell. The calculation of the scalar flux at P is described in detail in Reference 2 and will not be discussed herein. The following discussion gives a description of the ODD-K angular flux solution at the points A, B, C, and D in each outer surface mesh cell which are special output data required for use in the DAFT program. At each midpoint on the four surfaces of the mesh cell as shown in Figure 2, the discrete angular fluxes are solved for a hemisphere of the unit sphere centered about each point. The hemisphere about each point is subdivided into its four octants as shown in Figure 3. These octants represent an S_6 order angular quadrature. The numbers in the circles in Figure 3 represent points at



Figure 2. ODD-K R, Z Mesh Cell



Figure 3. ODD-K Discrete Directions

which the angular fluxes are obtained in the ODD-K S_6 solution. As shown in Figure 3, the unit vectors (μ,η,ξ) are represented in an S_6 angular quadrature as direction cosines $\mu_1, \mu_2, \mu_3, \eta_1, \eta_2, \eta_3, \xi_1, \xi_2, \xi_3$. With the same distribution of μ_i, η_i, ξ_i on each unit vector, the discrete directions on the surface of the hemisphere lie on latitudes and longitudes which maintain rotational symmetry in the hemisphere. This rotational symmetry is required in DAFT since all angular flux data are reduced to two separate angular flux distributions on the unit vectors μ and ξ.

The recommended discrete ordinate quadrature set for the R-Z geometry is the even moment symmetric set as suggested by Lathrop. The ODD-K quadrature sets including direction cosines and quadrature weights are presented in Table 1 for quadratures of order 2, 4, and 6. As indicated in Figure 3 and presented in Table 1, there are 30 discrete directions in an S_6 ODD-K solution. The numerical solution requires an initial direction solution for each ξ_i level (e.g., in S_6: $-\xi_1, -\xi_2, -\xi_3, +\xi_1, +\xi_2, +\xi_3$) in the hemisphere. These initial directions (indicated as 1, 8, 13, 16, 23, and 25 in Figure 3) are each assigned a quadrature weight of zero and do not enter into the scalar flux solution. Therefore, only 24 angular fluxes with non-zero weights are obtained in the S_6 hemisphere.

The angular flux data obtained from the ODD-K program for use in the DAFT code is the mesh cell surface data at the outer radius, top surface and bottom surface of the R-Z reactor geometry. These data, which are obtained as binary tape output, include the angular flux data at the points A for all the outer reactor radius mesh cells, C for all the top surface mesh cells, and D for all the bottom surface mesh cells.

The angular flux data obtained for each group, g, from the ODD-K program includes:

1. The bottom surface angular flux $B4_{ig}$ from each radial mesh interval, i, and group g.
2. The top surface angular flux $B6_{ig}$ for each radial mesh interval, i, and group g.
3. The outer radial surface angular flux, $B2_{jg}$ for each axial mesh interval j, and group g.

The angular flux data, $B4_{ig}$, $B6_{ig}$, $B2_{ig}$, are input to DAFT as a binary data tape. This tape contains the ODD-K problem title, the geometric data of mesh coordinate dimensions (i. e. , the radius or axial dimension of the surface which define the mesh cells in ODD-K), the quadrature direction cosines and weights, and the angular flux data. The first six logical tape records are:

1) The ODD-K problem title.
2) The radii of the mesh cell description (ODD-K input).
3) The axial dimensions of the mesh cell description (ODD-K input).
4) The direction cosines (ξ) of the ODD-K problem (M5 data list).
5) The direction cosines (μ) of the ODD-K problem (M7 data list).
6) The quadrature or direction weights of the ODD-K problem (W0 data list).

The data remaining on the tape consists of a set of logical tape records for each group in the multigroup solution. This tape is generated by ODD-K on the MSFC IBSYS version 13 tape number B-4. The actual data obtained from an ODD-K problem is nine logical records, of which, only three are actually needed for data processing in the DAFT program. The excess of data is obtained because the numerical solution of the group fluxes is solved in two passes (i. e. , downward and upward) through the mesh cell description; the angular fluxes, $B4_{ig}$ and $B6_{ig}$, are obtained each time the top (j=JM), and bottom (j=1), mesh cell rows are passed. The fourth logical record of the set of nine for each group contains the angular flux $B4_{ig}$, at the bottom row. The seventh logical record contains the angular flux, $B6_{ig}$, at the top row. The ninth logical record is the outer radial surface angular flux, $B2_{ig}$. The other records (first, second, third, fifth, sixth, and eighth) are excess data. The binary tape with nine records for each group, 1 to the number of groups in the ODD-K problem, is edited by the DAFT program.

TABLE 1

RECOMMENDED ODD-K QUADRATURE SETS

	Direction Cosines		Weights
	μ_m (M7)	ξ_m (M5)	W_m (W0)
	(S_2)		
1.	-1.000	-0.57735	0.0
2.	-0.57735	-0.57735	0.250
3.	+0.57735	-0.57735	0.250
4.	-1.0000	+0.57735	0.0
5.	-0.57735	+0.57735	0.250
6.	+0.57735	+0.57735	0.250
	μ_m (M7)	ξ_m (M5)	W_m (W0)
	(S_4)		
1.	-0.9367418	-0.3500212	0.0
2.	-0.8688903	-0.3500212	0.0833333
3.	-0.3500212	-0.3500212	0.0833333
4.	+0.3500212	-0.3500212	0.0833333
5.	+0.8688903	-0.3500212	0.0833333
6.	-0.4950046	-0.8688903	0.0
7.	-0.3500212	-0.8688903	0.0833333
8.	+0.3500212	-0.8688903	0.0833333
9.	-0.9367418	+0.3500212	0.0
10.	-0.8688903	+0.3500212	0.0833333
11.	-0.3500212	+0.3500212	0.0833333
12.	+0.3500212	+0.3500212	0.0833333
13.	+0.8688903	+0.3500212	0.0833333
14.	-0.4950046	+0.8688903	0.0
15.	-0.3500212	+0.8688903	0.0833333
16.	+0.3500212	+0.8688903	0.0833333
	μ_m (M7)	ξ_m (M5)	W_m (W0)
	(S_6)		
1.	-0.9637974	-0.2666355	0.0
2.	-0.9261808	-0.2666355	0.0440315
3.	-0.6815076	-0.2666355	0.0393017
4.	-0.2666355	-0.2666355	0.0440315
5.	+0.2666355	-0.2666355	0.0440315
6.	+0.6815076	-0.2666355	0.0393017
7.	+0.9261808	-0.2666355	0.0440315
8.	-0.7318110	-0.6815076	0.0
9.	-0.6815076	-0.6815076	0.0393017
10.	-0.2666355	-0.6815076	0.0393017
11.	+0.2666355	-0.6815076	0.0393017
12.	+0.6815076	-0.6815076	0.0393017
13.	-0.3770795	-0.9261808	0.0
14.	-0.2666355	-0.9261808	0.0440315
15.	+0.2666355	-0.9261808	0.0440315
16.	-0.9637974	+0.2666355	0.0
17.	-0.9261808	+0.2666355	0.0440315
18.	-0.6815076	+0.2666355	0.0393017
19.	-0.2666355	+0.2666355	0.0440315
20.	+0.2666355	+0.2666355	0.0440315
21.	+0.6815076	+0.2666355	0.0393017
22.	+0.9261808	+0.2666355	0.0440315
23.	-0.7318110	+0.6815076	0.0
24.	-0.6815076	+0.6815076	0.0393017
25.	-0.2666355	+0.6815076	0.0393017
26.	+0.2666355	+0.6815076	0.0393017
27.	+0.6815076	+0.6815076	0.0393017
28.	-0.3770795	+0.9261808	0.0
29.	-0.2666355	+0.9261808	0.0440315
30.	+0.2666355	+0.9261808	0.0440315

2.2 ANGULAR FLUX CALCULATIONS

The calculations performed in DAFT are restricted to the reduction of the outer surface angular flux data for each mesh cell into a limited number of surface area sources. The surface area of each outer mesh cell in the ODD-K problem is the initial data computed by DAFT. These data, obtained as a list of values, start at the bottom surface with the mesh cell at the centerline of the R, Z geometry. The surface mesh point index sequence is: one to the number of radial mesh cells IM; then IM + 1 to the sum of IM + JM, where JM is the number of axial mesh intervals; and then, IM + JM + 1 to IM + JM + IM. Therefore, the total number of source mesh point data from an ODD-K problem is IM + JM + IM sets of multigroup angular flux data. A schematic diagram of the DAFT surface area indexing system is shown in Figure 4.

The surface area of each bottom mesh cell follows as:

$$\Delta A_k = \pi(R^2_{i+1} - R^2_i)$$

where: k = the source mesh point index, k = 1 to IM

i = the radial mesh cell coordinate index, i=1 to IM

R_i = the radial mesh cell coordinate dimensions which define the coordinates of the mesh cell surfaces.

ΔA_k = the external surface area of kth surface mesh cell

The description of the surface area of each top mesh cell is identical to the bottom mesh cell as described above. The surface source mesh point index, k, is ordered such that: $\Delta A_{IM+JM+1} = \Delta A_{IM}$, $\Delta A_{IM+JM+2} = \Delta A_{IM-1}$, etc., to $\Delta A_{IM+JM+IM} = \Delta A_1$.

The surface area of each lateral surface mesh cell follows as:

$$\Delta A_k = 2\pi(R_{IM+1})(Z_{j+1} - Z_j)$$

where: k = the source mesh point index from IM+1 to IM+JM.

j = the axial mesh cell coordinate index, j=1 to JM.

Z_j = the axial mesh cell coordinate dimensions which define the axial coordinates of the mesh cell surfaces.



Figure 4. DAFT Surface Area Index

R_{IM+1} = the radius of the outer surface of the R, Z geometry.

ΔA_k = the external surface area of the kth surface mesh cell.

The areas are used in conjunction with the angular flux data to obtain integral surface source data for each incremental surface area.

Angular flux data calculations use the discrete direction quadrature weights and the conditions of a symmetric quadrature set to obtain the angular flux in a reduced form.

Consider the S_6 angular quadrature as illustrated in Figure 3. These data can be reduced to a polar distribution (the average angular flux at each ξ value) and an azimuthal distribution (the average angular flux at each μ value) on the unit sphere as shown in Figure 3. The polar distribution computation of surface angular flux follows as:

Average surface angular flux at each ξ value (latitude on the unit sphere)

$$N_{kg}^{l} = \frac{\sum_{m} N_{mkg} \times W0_m}{W^l} \qquad \text{for all values in level l}$$

where: m = the discrete direction index

N_{kg}^{l} = the average angular flux at each ξ value (latitude)

k = the source mesh interval index

N_{mkg} = the multigroup angular flux $B4_{mig}$, $B6_{mig}$, or $B2_{mjg}$

$W0_m$ = the quadrature weight (the area on the unit sphere)

W^l = the total area on the unit sphere for a given ξ value.

The DAFT program performs the above calculation in an S_6 angular quadrature with indices as follows:

for l = 1 , m = 29, 30

l = 2 , m = 24 - 27

l = 3 , m = 17 - 22

l = 4 , m = 2 - 7

l = 5 , m = 9 - 12

l = 6 , m = 14, 15

The average surface angular flux on each μ value (longitude on the unit sphere) is:

$$N^n_{kg} = \frac{\sum_m N_{mkg} W0_m}{W^n} \quad \text{for all m in level n}$$

where N^n_{kg} is the average angular flux at each μ value (longitude).

The DAFT program performs the above calculation in an S_6 angular quadrature with indices as follows:

for $n = 1$, $m = 2, 17$

$n = 2$, $m = 3, 9, 18, 24$

$n = 3$, $m = 4, 10, 14, 19, 25, 29$

$n = 4$, $m = 5, 11, 15, 20, 26, 30$

$n = 5$, $m = 6, 12, 21, 27$

$n = 6$, $m = 7, 22$

The angular flux data for each surface mesh cell are then used to calculate the surface source for each incremental area on the basis of input data which specifies the number of ODD-K surface mesh cells and the number of groups to be included in each surface area source in the DAFT calculations.

The following quantities are calculated for each surface area, s:

Total flux in surface area, s

$$N_s = \sum_{k=NCS_{s-1}}^{NCS_s} \sum_{l=1}^{6} \sum_{g=NGS}^{NGF} \Delta A_k N^l_{kg} W^l$$

where N_s = the total flux on the surface area, s, defined by the input list of data NCS_s, and the sum over neutron or photon energy groups, NGS to NGF.

NCS_s = the list of source mesh point index data defining the last source mesh point data to be included in the source, s. The first source mesh point data is the first surface area to be included in the source, S, or input value NCS_{s-1}.

s = the surface area source index number

NGS= the group number at which the summation over groups is to begin

NGF= the group number of the final group to be included in the summation over groups.

Energy Distribution in surface area, s

$$N_g = \sum_{k=NCS_{s-1}}^{NCS_s} \sum_{l=1}^{6} \Delta A_k N_{kg}^{l} W^{l}$$

Spatial Distribution in surface area, s

$$N_k = \left[\sum_{g=NGS}^{NGF} \sum_{l=1}^{6} \Delta A_k N_{kg}^{l} W^{l} \right] \Big/ \Delta A_k$$

Angular Distributions in surface area, s

Polar:

$$N_l = C \times \left[\sum_{g=NGS}^{NGF} \sum_{k=NCS_{s-1}}^{NCS_s} \Delta A_k N_{kg}^{l} \right]$$

Azimuthal:

$$N_n = C \times \left[\sum_{g=NGS}^{NGF} \sum_{k=NCS_{s-1}}^{NCS_s} \Delta A_k N_{kg}^{n} \right]$$

where the constant C is either 2.0 or 1.0 depending upon the surface (top, bottom or side) and the polar or azimuthal distribution. This factor accounts for the zero inward flux in the generalized (over the hemisphere of the unit sphere) integration in DAFT.

The data described above are obtained for surface areas defined by input data. Consider the source mesh interval indexing, k, which defines the bottom surface (k = 1,IM), the side surface (k = IM + 1 to IM + JM), and the top surface (k = IM + JM + 1 to IM + JM + IM) in sequence. There are IM + JM + IM sets of angular flux data for each range in the multigroup solution. The user specifies the number of surface area sources to be obtained from source interval data. Each surface area source includes a set of source mesh interval data. The surface area sources must be continuous from the first to the last source mesh interval to be coalesced into a surface area source. For example, if a DAFT problem has 105 source mesh intervals (36 radial and 33 axial) at the surface, the user may specify a DAFT problem with up to 105 surface area sources. If an 11 (input quantity NMAJOR) surface area source problem is desired, the user would choose 11 source mesh interval numbers such as 22, 26, 42, 47, 52, 58, 62, 69, 81, 100, 105 (input quantities NCS). The DAFT problem would calculate data from the sets of source mesh interval data as:

Surface Area, s	Source Mesh Intervals Included
1	1-22
2	23-36
3	37-42
4	43-47
5	48-52
6	53-58
7	59-62
8	63-69
9	70-81
10	82-100
11	101-105

This source mesh interval data would be obtained for a set of groups as specified by the input quantities NGS and NGF. A single DAFT problem calculates the spectral distribution which includes all groups from NGS to NGF.

Stacked DAFT problems can be run with different intervals of NCS, NGS, or NGF to obtain different spatial and/or energy distributed surface sources (e.g., fast and thermal neutron sources).

SECTION

3.0 PROGRAM LOGIC

The DAFT program is coded in FORTRAN-IV language for the IBM 7094 Model II computer. There are six subroutines in the DAFT program, and the function of each subroutine is given in Table 2. The structural composition of the DAFT program is shown in Figure 5. The flow of information within the DAFT program is given in Figure 6.

3.1 Tape Assignments

The DAFT program uses mnemonic designations for tape units. The use of mnemonic designations allows the unit buffers for those tape units not used by the program to be set to zero. The DAFT program deck listing in the Appendix, includes the required unit routine in the MAP machine assembly language. This routine sets the buffer lengths for tape units UN01, UN02, UN03, UN04, UN07, UN08 equal to zero. The use of the MAP routine allows the DAFT program to operate with 14,000 blank common storage locations in the FIOCS input/output package of the IBSYS Version 13 Monitor System.

The tape assignments in DAFT follow as:

Mnemonic Designation	Logical Tape No.	IBSYS MSFC Version 13 Tape No.	Description
MI	5	A-2	BCD Input
MO	6	B-1	BCD Output
MF	9	B-5	DAFT Processed Angular Flux Tape
MIF	11	B-6	ODD-K Angular Flux Tape

TABLE 2

CALCULATIONS PERFORMED BY DAFT SUBROUTINES

Subroutine	Calculation Performed
1. DAFT	a. Zero out all subscripted variables
	b. Read input data
	c. Control for two other subroutines
	d. Print problem input data
2. FLUX	a. Read ODD-K angular flux tape
	b. Place DAFT angular fluxes on tape
3. SCOUT	a. Print DAFT angular flux data
4. MAFIA	a. Print DAFT input data used in MAFIA
	b. Calculate distribution data
	c. Print calculated distribution data
5. AZMUT	a. Calculate average angular flux in the azimuthal angle
6. POLAR	a. Calculate average angular flux in the polar angle



Figure 5. DAFT Program Structure



Figure 6. Flow Chart for the DAFT Program

SECTION

4.0 INPUT DATA DESCRIPTION

Input data to the DAFT program includes: ODD-K problem size parameters, the source mesh interval indices describing the spatial integration, the group numbers to be used to obtain a sum over groups, and the binary data tape from an ODD-K neutron or photon energy problem. The program user may stack DAFT problems. If the program control data are input correctly, the ODD-K angular flux tape is only used in the first DAFT problem and all succeeding DAFT problems may use an intermediate binary data tape containing only the specific angular flux data used in DAFT.

4.1 CARD INPUT

The required data follow as:

Card Type	Variable	FORTRAN Format	Description
1	NSN	813	Angular quadrature order of the angular fluxes on the input binary tape (ODD-K) to be processed by DAFT (NSN = 6 for an S_6 ODD-K problem).
	NGP		Total number of groups of angular flux data on the input binary tape.
	NRI		Total number (IM) of radial mesh intervals in the ODD-K problem.
	NZI		Total number (JM) of axial mesh intervals in the ODD-K problem.
	NTYPE		Input binary tape control data. NTYPE = 2; the input tape is an ODD-K generated binary tape (MSFC IBSYS Version 13 tape no. B-6) NTYPE = 4: the input tape is a DAFT processed tape generated in a previous DAFT problem through the ODD-K tape. (MSFC IBSYS Version 13 tape no. B-5)

Card Type	Variable	FORTRAN Format	Description
	NMAJOR		Total number of surface area calculations to be performed in DAFT.
	NQUAD		Not required input.
	NOUT		Angular flux printed output control data. NOUT = 0: Do not print DAFT angular fluxes NOUT =1: Print angular fluxes for side, top and bottom surface for each group.
2	NCS	2413	The source mesh interval index number, k, which defines the last source mesh interval to be included in each surface area to be calculated in DAFT. The values of k must be in an increasing order. The program uses adjacent values of NCS to calculate each surface area data.
3	NGS	213	The group number in the multigroup data at which the group summation is to begin.
	NGF		The group number at which the group summation is to end.

4.2 TAPE INPUT

The required tape input follow as:

Tape B2, B4, B6 (Binary Tape) — The angular flux data as binary tape data is required in the tape format as described in Section 2.1. The user provides a tape on either B-5 or B-6 depending on the value of the input quantity NTYPE. The ODD-K tape format is described in Section 2.1 and the DAFT binary flux tape is described in Section 5.0.

SECTION

5.0 OUTPUT DATA DESCRIPTION

The DAFT program prints all input data and computed data of a single problem. The first line of each DAFT problem output is the variable dimension statement, "YOUR DAFT PROBLEM HAS USED N LOCATIONS OF THE AVAILABLE M REAL NUMBER STORAGE LOCATIONS." This line indicates the overall size of a DAFT problem. The DAFT program will terminate all succeeding DAFT problems if $N > M$.

The second set of data is obtained only for a problem which employs an input ODD-K binary tape. The printed output is the ODD-K problem title, and the lists of data (R, Z, M5, M7, and W 0). Also, the angular flux data B2, B4, B6 can be printed (if NOUT = 1) as a column for each surface mesh cell and a line for each discrete direction (e.g., directions 1 through 30 for an S_6 ODD-K problem).

The third set of data is the surface area of each source mesh interval from 1 to IM + JM + IM (DAFT values: NRI + NZI + NRI).

The fourth set of printed output is the quantities NRI, NZI, NSN, NGP, and NMAJOR.

The fifth set of printed data is the surface area data for each surface area 1 to NMAJOR. The data are: the surface area index, s, surface mesh cell indices, NCS_{s-1} and NCS_s; the total flux N_g, the energy distribution N_g (a histogram with NGS to NGF values); the spatial distribution N_l for source mesh points NCS_{s-1} to NCS_s; and the angular distributions N_{ls}, N_{ns} with the first column, the polar, and the second column, the azimuthal distribution.

APPENDIX

THE FORTRAN SOURCE DECK LISTING IS PRESENTED ON PAGES 28 THROUGH 44

```
$IBFTC DAFFT   LIST,REF,DECK,NODD,M94/2,XR7
C
C
C     DAFT CODE - DDK ANGULAR FLUX TAPE CODE TO REARRANGE THE SN
C                 TRANSPORT ANGULAR FLUXES FOR THE MAP OR MAFIA SUBROUT.
C
      DIMENSION TITLE(12)
      DIMENSION X(14000)
      DIMENSION IX(100)
      DIMENSION GA(12,12),GW(12,12),ISURF(3)
      DIMENSION NGOOD(100)
C
      COMMON NSN   ,NGP   ,NRI   ,NZI   ,NDM   ,NDM2  ,NTYPE ,NRM   ,
     1       NZM   ,NOUT  ,NMAJOR,NMAXI ,MI    ,MO    ,MIF   ,MF    ,
     2       MSF   ,NQUAD ,MTF   ,NDET  ,ISURF ,NGPTOT,NGOOD
C
      NMAXRS = 14000
      MI = 5
      MO = 6
      MF = 9
      MSF = 10
      MIF = 11
      MTF = 12
      REWIND MF
      REWIND MSF
      REWIND MIF
      REWIND MTF
C
    5 READ(MI,105)NSN,NGP,NRI,NZI,NTYPE,NMAJOR,NQUAD,NOUT,NGPTOT
  105 FORMAT(24I3)
      READ(MI,105) (NGOOD(K),K=1,NGPTOT)
   10 DO 15 I = 1,NMAXRS
   15 X(I) = 0.0
      NRM = NRI + 1
      NZM = NZI + 1
      I1 = NRI
      I2 = NZI
      I3 = NRI + 1
      I4 = NZI + 1
      I5 =(NSN+4)*NSN/2
      NDM=I5
      I6 = I5 / 2
      NDM2=I6
      I7 = I5 * I1
      I8=MAX0(I1,I2)*I5
      J1 = 1
```

```
C R(NR1+1)
      J2 = J1 + I3
C Z(NZI+1)
      J3 = J2 + I4
C RI(NRI)
      J4 = J3 + I1
C ZI(NZI)
      J5 = J4 + I2
C M5(NDM)
      J6=J5+I5
C M7(NDM)
      J7=J6+I5
C W0(NDM)
      J8=J7+I5
C
C     TEST TO SEE IF A DAFT ANGULAR FLUX
C     TAPE IS AVAILABLE ON TAPE9,A-5,
C     IF SO, TRANSFER IMMEDIATELY TO MAP OR MAFIA SUBROUTINES
C     NTYPE = 1/2/3/4, B5/B5/A5/A5, FOR MAP/MAFIA/MAP/MAFIA
C
      IF(NTYPE.GT.2) GO TO 20
C N2(NZI,NDM)
      J9=J8+I2*I5
C N4(NRI,NDM,2)
      J10= J9 + I1*I5*2
C B2(NZI,NDM)
      J11=J10+I2*I5
C B4(NRI,NDM)
      J12=J11+I1*I5
C B6(NRI,NDM)
      J13=J12+I1*I5
C D(NRI*NDM OR NZI*NDM)
      J14=J13+I8
C FLUXA(NZI)
      J15=J14+I2
C FLUXB(NRI)
      J16=J15+I1
C FLUXC(NRI)
      J17=J16+I1
      NSTORE = NMAXRS - J17
      WRITE(MO,100)J17,NMAXRS
      IF(NSTORE.GE.0) GO TO 90
  100 FORMAT(1HO,5X,26HYOUR DAFT PROBLEM HAS USED I6,27H LOCATIONS OF TH
     1E AVAILABLE I6,31H REAL NUMBER STORAGE LOCATIONS )
      IF(NSTORE.LT.0) CALL EXIT
C
```

```
C
   90 CALL FLUX(X(J1 ),X(J2 ),X(J3 ),X(J4 ),X(J5 ),X(J6 ),X(J7 ),X(J8 ),
     1          X(J9 ),X(J10),X(J11),X(J12),X(J13),X(J14),X(J15),X(J16),
     2          I1,I2,I3,I4,I5,I6,I7,I8)
   20 IF(NTYPE.EQ.2.OR.NTYPE.EQ.4) GO TO 89
      NDET = NMAJOR
      I8 = 2*NRI+NZI
      I9 = NQUAD/2
      I10 = NGP
      I11 = NSN
      I12 = NDET
C PHIA(NRI+NZI+NRI,NDM)
      J9 = J8 + I8*I5
C PHIT(4,NGP,NDET)
      J10 = J9 + 4*I10*I12
C RD(NDET)
      J11 = J10 + I12
C ZD(NDET)
      J12 = J11 + I12
C THEI(NQ)
      J13 = J12 + I9
C THEG(NQ)
      J14 = J13 + I9
C COST(NQ,NRI+NRI+NZI)
      J15 = J14 + I9*I8
C COSP(NQ,NRI+NRI+NZI)
      J16 = J15 + I9*I8
C RHO2(NQ,NRI+NRI+NZI)
      J17 = J16 + I9*I8
C PP(NSN,NRI+NRI+NZI)
      J18 = J17 + I11*I8
C PU(NSN,NRI+NRI+NZI)
      J19 = J18 + I11*I8
C PD(NSN,NRI+NRI+NZI)
      J20 = J19 + I11*I8
C TA(NSN)
      J21 = J20 + I11
C PA(NSN)
      J22 = J21 + I11
C RESP(NGP,10)
      J23 = J22 + I10*10
C NAME(NDET)
      J24 = J23 + I12
C WT(NSN)
      J25 = J24 + I11
C WTA(NSN)
```

```
      J26 = J25 + I11
      NSTORE = NMAXRS - J26
      IF(NSTORE.GE.0) GO TO 88
      WRITE(MO,222) J22,NMAXRS
  222 FORMAT(1HO,5X,26HYOUR MAP  PROBLEM HAS USED I6,27H LOCATIONS OF TH
     1E AVAILABLE I6,31H REAL NUMBER STORAGE LOCATIONS )
      CALL EXIT
C
   88 CALL MAP(X(J1 ),X(J2 ),X(J3 ),X(J4 ),X(J5 ),X(J6 ),X(J7 ),X(J8 ),
     1          X(J9 ),X(J10),X(J11),X(J12),X(J13),X(J14),X(J15),X(J16),
     2          X(J17),X(J18),X(J19),X(J20),X(J21),X(J22),X(J23),X(J24),
     3          X(J25),
     4          I1,I2,I3,I4,I5,I6,I7,I8,I9,I10,I11,I12)
      GO TO 5
C
   89 I8 = 2*NRI+NZI
      I9 = NMAJOR
      I10= NMAXI
      I11 = NGP
      I12 = NSN
C PHIA(NRI+NZI+NRI,NDM)
      J9 = J8 + I8 * I5
C PHIG(NGP,NMAJOR)
      J10 = J9 + I11*I9
C PHIS(NRI+NZI+NRI,NMAJOR
      J11 = J10 + I8 *I9
C PHIP(NSN,NMAJOR)
      J12 = J11 + I12*I9
C PHIL(NSN,NMAJOR)
      J13 = J12 + I12*I9
C PHIT(NMAJOR)
      J14 = J13 + I9
C PA(NSN)
      J15 = J14 + I12
C PB(NSN)
      J16 = J15 + I12
C WTP(NSN)                                                          9
      J17 = J16 + I12
C WTA(NSN)                                                          10
      J18 = J17 + I12
C AREA(I8)
      J19=J18+I8
C NCS(NMAJOR)                                                       11
      J20=1
      NSTORE = NMAXRS - J19
      WRITE(MO,323) J19,NMAXRS
```

```
  323 FORMAT(1H1,5X,26HYOUR DAFT PROBLEM HAS USED I6,27H LOCATIONS OF TH
     1E AVAILABLE I6,31H REAL NUMBER STORAGE LOCATIONS )
      IF(NSTORE.LT.0) CALL EXIT
C
   11 CALL MAFIA(X(J1 ),X(J2 ),X(J3 ),X(J4 ),X(J5 ),X(J6 ),X(J7 ),
     1           X(J8 ),X(J9 ),X(J10),X(J11),X(J12),X(J13),X(J14),
     2           X(J15),X(J16),X(J17),X(J18),
     3           IX(J20),
     4           I1,I2,I3,I4,I5,I6,I7,I8,I9,I10,I11,I12)
      GO TO 5
      END
```

```
$IBFTC FLUXS   LIST,REF,DECK,NODD,M94/2,XR7
      SUBROUTINE FLUX(R,Z,RI,ZI,M5,M7,WO,N2,N4,B2,B4,B6,D,FLUXA,FLUXB,
     1FLUXC,I1,I2,I3,I4,I5,I6,I7,I8)
C
      DIMENSION R(I3),Z(I4),RI(I1),ZI(I2),M5(I5),M7(I5),WO(I5),N2(I2,I5)
     1,N4(I1,I5,2),B2(I2,I5),B4(I1,I5),B6(I1,I5),D(I8)   ,FLUXA(I2),FLUX
     2B(I1),FLUXC(I1),
     3TITLE(12)
      DIMENSION GA(12,12),GW(12,12),ISURF(3)
      DIMENSION NGOOD(100)
C
      COMMON NSN   ,NGP   ,NRI   ,NZI   ,NDM   ,NDM2  ,NTYPE ,NRM   ,
     1       NZM   ,NOUT  ,NMAJOR,NMAXI ,MI    ,MO    ,MIF   ,MF    ,
     2       MSF   ,NQUAD ,MTF   ,NDET  ,ISURF ,NGPTOT,NGOOD
C
      REAL M5,M7,N2,N4
C
C     TAPE UNIT 8-5 HAS THE FOLLOWING ODDK PROBLEM DATA FOLLOWED BY
C     ANGULAR FLUX ARRAYS
C     ODDK NAME         DESCRIPTION                DIMENSION        RECORD
C     TITLE              R*    900                                      1
C     A04                QUADRATURE ORDER                                2
C     G01                NO. OF GROUPS
C     I02                NO. OF RADIAL MESH INT.
C     I03                NO. OF AXIAL MESH INT.
C     R0                 RADII                     (I02)                3
C     Z0                 AXIAL DIMENSIONS          (I03)                4
C     M5                 DIRECTION COSINES         (A04*(A04+4)/2)      5
C     M7                 DIRECTION COSINES         (A04*(A04+4)/2)      6
C     W0                 POINT WEIGHTS             (A04*(A04+4)/2)      7
C     THE ANGULAR FLUXES FOLLOW.
C
C     EACH ARRAY IS OF THE FOLLOWING DIMENSIONS AND IN THE ORDER SHOWN
C     ODDK NAME  DESCRIPTION                       DIMENSION
C      B4          TOP FLUX FOR EACH SLICE        I02 * (A04*(A04+4)/2) 81
C      B6          BOTTOM FLUX FOR EACH SLICE     I02 * (A04*(A04+4)/2) 82
C      N4          CENTERED FLUX FOR EACH SLICE   I02 * (A04*(A04+4)/2) 83
C
C      THE ABOVE THREE QUANITIES ARE OBTAINED ON A DOWN AND UP PASS
C      HENCE,THE DOWN PASS GIVES B6 AND UP PASS B4
C      IF LAST SLICE ON UPWARD PASS,
C      B2          RIGHT FLUX FOR ALL SLICES      I03 * (A04*(A04+4)/2)  9
C      REPEAT FROM B4 FOR EACH GROUP
C
C
C
```

```
C     READ DATA FROM DDK TAPE, MIF
C
      READ(MIF)(TITLE(I),I=1,12)
      WRITE (MO,200) (TITLE(I),I=1,12)
  200 FORMAT(1H1,15X,12A6,/)
      WRITE (MO,210) NSN,NGP,NRI,NZI
  210 FORMAT(25X,41H ORDER OF SN APPROXIMATION - - - - - - - I3,/,
     1       25X,41H NUMBER OF ENERGY GROUPS - - - - - - - - I3,/,
     2       25X,41H NUMBER OF RADIAL MESH INTERVALS - - - - I3,/,
     3       25X,41H NUMBER OF AXIAL MESH INTERVALS  - - - - I3,/)
      NRM = NRI + 1
      NZM = NZI + 1
      READ(MIF)(R(N),N=1,NRM)
      WRITE (MO,215)
  215 FORMAT(1HO,20X,18H RADIAL DIMENSIONS )
      WRITE (MO,220)(R(N),N=1,NRM)
  220 FORMAT(// ,8(1X,1PE12.5))
      READ(MIF)(Z(N),N=1,NZM)
      WRITE (MO,225)
  225 FORMAT(1HO,20X,17H AXIAL DIMENSIONS )
      WRITE (MO,220) (Z(N),N=1,NZM)
      NDM = NSN*(NSN+4)/2
      NDM2= NDM / 2
      READ(MIF)(M5(N),N=1,NDM)
      WRITE (MO,230)
  230 FORMAT(1HO,20X,18H DIRECTION COSINES )
      WRITE (MO,220) (M5(N),N=1,NDM)
      READ(MIF)(M7(N),N=1,NDM)
      WRITE (MO,235)
  235 FORMAT(1HO,20X,24H MORE DIRECTION COSINES )
      WRITE (MO,220) (M7(N),N=1,NDM)
      READ(MIF)(WO(N),N=1,NDM)
      WRITE (MO,240)
  240 FORMAT(1HO,20X,14H POINT WEIGHTS )
      WRITE (MO,220) (WO(N),N=1,NDM)
      DO 14 I=1,NRI
   14 RI(I) = (R(I)+R(I+1))/2.
      DO 16 J=1,NZI
   16 ZI(J) = (Z(J)+Z(J+1))/2.
CCCCC DO 1000 K=1,NGP
      DO 1000 K=1,NGPTOT
      NTOT = NRI*NDM
C
C     EXTRACT BOTTOM SURFACE ANGULAR FLUXES,B6
C
      READ(MIF)(D(I),I=1,NTOT)
```

```
      READ(MIF)(D(I),I=1,NTOT)
      READ(MIF)(D(I),I=1,NTOT)
      READ(MIF)(D(I),I=1,NTOT)
      DO 27 M=1,NDM2
      DO 27 I=1,NRI
      MA = I + (M-1)*NRI
   27 B6(I,M)=D(MA)
      READ(MIF)(D(I),I=1,NTOT)
      READ(MIF)(D(I),I=1,NTOT)
      READ(MIF)(D(I),I=1,NTOT)
C
C     EXTRACT TOP SURFACE ANGULAR FLUXES,B4
C
      NDM3 = NDM2+1
      DO 78 M=NDM3,NDM
      DO 78 I=1,NRI
      MA = I+(M-1)*NRI
   78 B4(I,M)=D(MA)
      READ(MIF)(D(I),I=1,NTOT)
      NTOT = NZI*NDM
      READ(MIF)(D(I),I=1,NTOT)
C
C     EXTRACT SIDE SURFACE ANGULAR FLUXES,B2
C
      DO 99 M=1,NDM
      DO 99 J=1,NZI
      MA = J+(M-1)*NZI
   99 B2(J,M) = D(MA)
      IF (K.NE.1) GO TO 101
C
C     WRITE GEOMETRIC,ANGULAR QUADRATURE,AND ANGULAR FLUX DATA ON A5,9
C
      WRITE (MF) (R(I),I=1,NRM),
     1           (Z(J),J=1,NZM),
     2           (RI(I),I=1,NRI),
     3           (ZI(J),J=1,NZI),
     4           (M5(M),M=1,NDM),
     5           (M7(M),M=1,NDM),
     6           (WO(M),M=1,NDM)
  101 IF(NGOOD(K).LE.0) GO TO 1000
C 101 WRITE (MF)((B2(J,M),J=1,NZI),M=1,NDM)
      WRITE (MF)((B2(J,M),J=1,NZI),M=1,NDM)
      WRITE(MF) ((B4(J,M),J=1,NRI),M=1,NDM),
     1          ((B6(J,M),J=1,NRI),M=1,NDM)
C
C     PRINT ANGULAR FLUX DATA
```

```
C
      IF(NOUT.LE.0) GO TO 1000
      CALL SCOUT(B2, ZI, M5, NZI, NDM, NZI, NDM,K,18H SIDE ANGULAR FLUX)
      CALL SCOUT(B4, RI, M7,NRI,NDM,NRI,NDM, K, 18H TOP  ANGULAR FLUX)
      CALL SCOUT(B6, RI, M7,NRI,NDM,NRI,NDM, K, 18H BOT.ANGULAR FLUX )
 1000 CONTINUE
      REWIND MIF
      REWIND MF
      RETURN
      END
```

```
$IBFTC SCOUTT  LIST,REF,DECK,NODD,M94/2,XR7
      SUBROUTINE SCOUT(SC,RI,ZI,I1,I2,I3,I4,K,NAME)
      DIMENSION SC(I1,I2),RI(I3),ZI(I4),NAME(3)
      WRITE (6,12345)
12345 FORMAT(1HO,30X,6HSCOUT )
      CALL TIMEX
      WRITE (6,1260) (NAME(I),I=1,3),K
      DO 210 ISTAR = 1,I4,8
      ISTOP=ISTAR + 7
      ISTOP = MIN0(I4,ISTOP)
      WRITE(6,1175)(II,II=ISTAR,ISTOP)
      WRITE(6,1171) (ZI(II),II=ISTAR,ISTOP)
      DO 210 J=1,I3
  210 WRITE(6,1170) J,RI(J),(SC(J,I),I=ISTAR,ISTOP)
 1170 FORMAT(I4,F8.2,1P8E12.4)
 1171 FORMAT(12X,8(2X,F8.2,2X))
 1175 FORMAT(4HOI/J,8X,8(I6,6X))
 1260 FORMAT(1H1,3A6,11H GROUP NO.=,I6)
      RETURN
      END
```

```
$ORIGIN         ALPHA,SYSUT3,REW
$IBFTC MAFI     LIST,REF,DECK,NODD,M94/2,XR7
      SUBROUTINE MAFIA(R,Z,RI,ZI,M5,M7,WO,PHIA,PHIG,PHIS,PHIP,PHIL,PHIT,
     1PA,PB,WTP,WTA,AREA,NCS,
     2                 I1,I2,I3,I4,I5,I6,I7,I8,I9,I10,I11,I12)
C
      DIMENSION R(I3),Z(I4),RI(I1),ZI(I2),M5(I5),M7(I5),WO(I5),PHIA(I8,
     1I5),PHIG(I11,I9),PHIS(I8 ,I9),PHIP(I12,I9),PHIL(I12,I9),PHIT(I9),
     2PA(I12),PB(I12),WTP(I12),WTA(I12),AREA(I8),NCS(I9),CAS(8)
      DIMENSION GA(12,12),GW(12,12),ISURF(3)
      DIMENSION NGOOD(100)
C
      COMMON NSN    ,NGP    ,NRI    ,NZI    ,NDM    ,NDM2   ,NTYPE ,NRM    ,
     1       NZM    ,NOUT   ,NMAJOR,NMAXI  ,MI     ,MO     ,MIF   ,MF     ,
     2       MSF    ,NQUAD  ,MTF    ,NDET   ,ISURF  ,NGPTOT,NGOOD
C
      REAL M5,M7
C
C PHIL(L,N)  IS THE AVERAGE FLUX IN EACH AZMUTHIAL BAND L FOR EACH MAJOR
C INTERVAL
C PHIP(L,N)  IS THE AVERAGE FLUX IN EACH POLAR BAND L FOR EACH MAJOR
C INTERVAL
C PHIS(I,N) IS THE SPATIAL DISTRIBUTION IN THE MAJOR INTERVAL N
C PHIT(N) IS THE TOTAL FLUX IN THE MAJOR INTERVAL N
C PHIG(K,N) IS THE SPECTRUM IN THE MAJOR INTERVAL N
C WTP(L)IS THE AREA ON THE UNIT SPHERE FOR  EACH BAND L IN THE POLAR
C DIRECTION
C WTA(L)IS THE AREA ON THE UNIT SPHERE FOR EACH BAND L IN THE AZMUTHIAL
C DIRECTION
C
      READ (MF) (R(I),I=1,NRM),
     1          (Z(J),J=1,NZM),
     2          (RI(I),I=1,NRI),
     3          (ZI(J),J=1,NZI),
     4          (M5(L),L=1,NDM),
     5          (M7(L),L=1,NDM),
     6          (WO(L),L=1,NDM)
      READ(MI,200)(NCS(N),N=1,NMAJOR)
      READ (MI,200)NGS,NGF
  200 FORMAT(24I3)
      PI = 3.14159286
      DO 2 I=1,8
    2 CAS(I) = 0.
      DO 8 I = 1,NMAJOR
      PHIT(I) = 0.0
      DO 6 L=1,I8
```

```
    6 PHIS(L,I)=0.0
      DO 7 L=1,NSN
      PHIP(L,I) = 0.0
    7 PHIL(L,I) = 0.0
      DO 8 K = 1,NGP
    8 PHIG(K,I) = 0.0
      N3 = NRI + NZI
      N4 = N3 + NRI
      DO 9 I = 1,N4
      J=I-NRI
      IF(I.LE.NRI) AREA(I) = PI*(R(I+1)*R(I+1)-R(I)*R(I))
      IF(I.GT.NRI.AND.I.LE.N3) AREA(I) = 2.0*PI*R(NRI+1)*(Z(J+1)-Z(J))
      IF(I.LE.N3) GO TO 9
      J = N4 - I + 1
      AREA(I) = AREA(J)
    9 CONTINUE
      WRITE (MO,897)(I,AREA(I),I=1,N4)
  897 FORMAT(24H1AREAS OF EACH INTERVAL  //(5X,I5,1PE15.5))
      DO 10 K = 1,NGP
      N1 = NRI + 1
      N2 = NRI + NZI
      READ (MF) ((PHIA(I,M),I=N1,N2),M=1,NDM)
      N1 = NRI + NZI + 1
      N2 = N4
      READ (MF) ((PHIA(I,M),I=N1,N2),M=1,NDM),((PHIA(I,M),I=1,NRI),M=1,
     1NDM)
      IF(K.LT.NGS.OR.K.GT.NGF) GO TO 10
      NS = 1
      DO 15 N = 1,NMAJOR
      NF = NCS(N)
      DO 20 I = NS,NF
C
C
      CALL AZMUT(NSN,N4,NDM,NDM2,PHIA,WTA,PB,WO,I)
C
      CALL POLAR(NSN,N4,NDM,NDM2,PHIA,WTP,PA,WO,I)
C
C
      TEM1 = 0.0
      TEM2 = 0.0
      DO 25 L = 1,NSN
      TEM1 = TEM1 + PA(L)*WTP(L)
      PHIP(L,N) = PHIP(L,N) + PA(L)*AREA(I)
      PHIL(L,N) = PHIL(L,N) + PB(L)*AREA(I)
   25 TEM2 = TEM2 + PB(L)*WTA(L)
      TEM3 = TEM1 * AREA(I)
```

```
      PHIG(K,N) = PHIG(K,N) + TEM3
      PHIT(N) = PHIT(N) + TEM3
   20 PHIS(I,N) = PHIS(I,N) + TEM3
   15 NS=NCS(N)+1
   10 CONTINUE
      REWIND MF
      WRITE(MO,100) NRI,NZI,NSN,NGP,NMAJOR
  100 FORMAT(55H1    MONTE CARLO ANGULAR FLUX DATA FROM TRANSPORT CALC. /
     1/       5X,46H TRANSPORT MODEL.....RADIAL INTERVALS........ I6   /
     1/       5X,46H                .....AXIAL INTERVALS........ I6   /
     1/       5X,46H                .....ANGULAR QUADRATURE..... I6   /
     1/       5X,46H                .....NO. OF GROUPS.......... I6   /
     1/       5X,46H MONTE CARLO MODEL....NO. OF SOURCE INTERVALS. I6   )
      NS = 1
      DO 30 K= 1,NMAJOR
      NF = NCS(K)
      WRITE(MO,105) K,NS,NF
  105 FORMAT(1H1,6X,28H SOURCE INTERVAL NO......... I6  /
     1/       7X,28H   INTERVAL COLLAPSE FROM... I6 /
     1/       7X,28H                     TO..... I6 //)
      WRITE(MO,110) PHIT(K)
  110 FORMAT(9X,28H  TOTAL FLUX.............. 1PE15.5 /)
      WRITE(MO,114)
      WRITE(MO,115) (N,PHIG(N,K),N=1,NGP)
  114 FORMAT(9X,24H  SPECTRAL DISTRIBUTION  /
     1       9X,22H    GROUP NO.    FLUX   /)
  115 FORMAT(12X,I6,4X,1PE15.5)
      DO 118 N=NS,NF
  118 PHIS(N,K)=PHIS(N,K)/AREA(N)
      WRITE(MO,116)
      WRITE(MO,115) (N,PHIS(N,K),N=NS,NF)
  116 FORMAT(9X,24H  SPATIAL DISTRIBUTION   /
     1       9X,22H    INTERVAL     FLUX   /)
      WRITE(MO,119)
      N2 = NRI + NZI
      IF(NF.LE.NRI.OR.NS.GT.N2) GO TO 120
      DO 121 L=1,NSN
  121 PHIP(L,K) = PHIP(L,K)*2.0
      GO TO 122
  120 DO 123 L= 1,NSN
  123 PHIL(L,K) = PHIL(L,K)*2.0
  122 WRITE(MO,117) (L,CAS(L),PHIP(L,K),PHIL(L,K),L=1,NSN)
  119 FORMAT(9X,24H  ANGULAR DISTRIBUTION   /
     1       9X,50H    LEVEL   COSINE      FLUX(PHI)     FLUX(THETA) /)
  117 FORMAT(12X,I6,4X,1P3E15.5)
   30 NS=NCS(K)+1
```

RETURN
END

```
$IBFTC POLA    LIST,REF,DECK,NODD,M94/2,XR7
      SUBROUTINE POLAR(NSN,N4,NDM,NDM2,PHIA,WTP,PA,WO,I)
C
C **** THIS ROUTINE CALCULATES THE AVERAGE FLUX IN EACH OF THE BANDS DEF
C INING THE POLAR DISTRIBUTION AT THE MESH POINT I. VALUES 1-NI ARE IN
C THE DOWNWARD DIRECTED HEMISPHERE AND VALUES NI+1-NSN ARE IN THE UPWARD
C DIRECTED HEMISPHERE
C
      DIMENSION PHIA(N4,NDM),WTP(NSN),PA(NSN),WO(NDM)
      DIMENSION GA(12,12),GW(12,12),ISURF(3)
      DIMENSION NGOOD(100)
C
      COMMON NSN    ,NGP    ,NRI    ,NZI    ,NDM    ,NDM2   ,NTYPE ,NRM    ,
     1       NZM    ,NOUT   ,NMAJOR,NMAXI  ,MI     ,MO     ,MIF   ,MF     ,
     2       MSF    ,NQUAD  ,MTF    ,NDET   ,ISURF  ,NGPTOT,NGOOD
C
      LS = NDM2+2
      LR = 2
      LI = 3
      LU = 4
      NI = NSN/2
      LT = NSN
      DO 10 L = 1,NI
      PA(LI) = 0.
      PA(LU) = 0.
      WTP(LI)=0.0
      WTP(LU) = 0.0
      DO 20 M = 1,LT
      PA(LI) = PA(LI) + PHIA(I,LS)*WO(LS)
      PA(LU) = PA(LU) + PHIA(I,LR)*WO(LR)
      WTP(LI) = WTP(LI) + WO(LR)
      WTP(LU) = WTP(LU) + WO(LS)
      LR = LR + 1
   20 LS = LS + 1
      PA(LI)= PA(LI)/WTP(LI)
      PA(LU)= PA(LU)/WTP(LU)
      LT = LT - 2
      LI = LI - 1
      LU = LU + 1
      LR = LR + 1
   10 LS = LS + 1
      RETURN
      END
```

```
$IBFTC AZMUTH  LIST,REF,DECK,NODD,M94/2,XR7
      SUBROUTINE AZMUT(NSN,N4,NDM,NDM2,PHIA,WTA,PB,WO,I)
C
C **** THIS ROUTINE CALCULATES THE AVERAGE FLUX IN EACH OF THE BANDS DEF
C INING THE AZMUTHIAL DISTRIBUTION AT THE MESH POINT I. VALUES 1-NI ARE
C IN THE INWARD DIRECTED HEMISPHERE AND VALUES NI+1-NSN ARE IN THE OUTWA
C RD DIRECTED HEMISPHERE
C
      DIMENSION PHIA(N4,NDM),WTA(NSN),PB(NSN),WO(NDM)
      DIMENSION GA(12,12),GW(12,12),ISURF(3)
      DIMENSION NGOOD(100)
C
      COMMON NSN   ,NGP   ,NRI   ,NZI   ,NDM   ,NDM2  ,NTYPE ,NRM   ,
     1       NZM   ,NOUT  ,NMAJOR,NMAXI ,MI    ,MO    ,MIF   ,MF    ,
     2       MSF   ,NQUAD ,MTF   ,NDET  ,ISURF ,NGPTOT,NGOOD
C
      NSN2=NSN/2
      NI = 1
      DO 10 L = 1,NSN
      ND = L + 1
      NU = ND + NDM2
      PB(L) = 0.
      WTA(L)=0.0
      NIN = NSN
      DO 20 K = 1,NI
      PB(L) = PB(L) + PHIA(I,ND)*WO(ND)
      PB(L) = PB(L) + PHIA(I,NU)*WO(NU)
      WTA(L) = WTA(L) + WO(ND)
      WTA(L) = WTA(L) + WO(NU)
      ND = ND + NIN
      NU=NU+NIN
   20 NIN = NIN - 2
      PB(L) = PB(L)/WTA(L)
      IF(L.GT.NSN2) NI = NI -1
      IF(L.LT.NSN2) NI = NI +1
   10 CONTINUE
      RETURN
      END
```

```
$IBFTC MAPMSF  LIST,REF,DECK,NODD,M94/2,XR7
       SUBROUTINE MAP(R,Z,RI,ZI,M5,M7,WO,PHIA,PHIT,RD,ZD,THEI,THEG,COST,
      1                COSP,RHO2,PP,PU,PD,TA,PA,RESP,NAME,WT,WTA,
      2                I1,I2,I3,I4,I5,I6,I7,I8,I9,I10,I11,I12)
       RETURN
       END
```